

Excel. Использование ВПР для решения общих проблем

Это четвертая глава книги Билла Джелена. [Всё о ВПР: от первого применения до экспертного уровня.](#)

Есть ли этот товар в основном списке? Проблема: у меня есть список товаров, заказанных сегодня. Мне нужно сравнить его со списком позиций, которые есть в наличии на складе.

	A	B	C	D
1	Товары в заказе	В наличии?		
2	BR15-3			Товары на складе
3	ER49-20			BG33-17
4	BG33-17			BG33-3
5	BR15-3			BR15-3
6	CR50-6			CR50-2
7				Cross50-5
8				ER 46-33
9				ER41-4
10				ER46-28
11				ER46-7
12				ER49-21
13				ER80-63
14				P411A
15				RG75-3
16				W25-6
17				

Рис. 4.1. Какие элементы из столбца С есть в наличии на складе (в столбце D)?

Стратегия: введите в ячейку B2 формулу ВПР с использованием таблицы подстановки D2:D16. Для номера столбца подстановки, используйте число 1. Если элемент найден в списке товаров на складе, ВПР вернет название товара. Если товара нет в наличии, вы получите ошибку #Н/Д (рис. 4.2).

Формула будет выглядеть так =ВПР(A2; =\$D\$2:\$D\$16;1;ЛОЖЬ) или так =ВПР(A2;Таблица24[Все];Товары на складе];1;ЛОЖЬ)

	A	B	C	D	E	F
1	Товары в заказе	В наличии?				
2	BR15-3	BR15-3		Товары на складе		
3	ER49-20	#Н/Д		BG33-17		
4	BG33-17	BG33-17		BG33-3		
5	BR15-3	BR15-3		BR15-3		
6	CR50-6	#Н/Д		CR50-2		
7				Cross50-5		
8				ER 46-33		
9				ER41-4		
10				ER46-28		
11				ER46-7		
12				ER49-21		
13				ER80-63		
14				P411A		
15				RG75-3		
16				W25-6		
17						

Рис. 4.2. Товары с #Н/Д отсутствуют на складе

Если вы выполняете анализ для себя, то вы вполне можете удовлетвориться таким результатом. Можно отсортировать список A1:B6 по столбцу B по возрастанию. Все имеющиеся в наличии товары будут отображаться в верхней части списка (рис. 4.3).

	A	B	C
1	Товары в заказе	В наличии?	
2	BR15-3	BR15-3	
3	BG33-17	BG33-17	
4	BR15-3	BR15-3	
5	ER49-20	#Н/Д	
6	CR50-6	#Н/Д	
7			

Рис. 4.3. Товары, которых нет в наличии отсортированы вниз таблицы

Если вы должны предоставить этот список вашему придирчивому менеджеру, он может не понять, что #Н/Д означает отсутствие товара на складе. Вы можете улучшить формулу, используя функции ЕСЛИОШИБКА, ЕОШИБКА и ЕСЛИ. Ниже показаны пять разных способов улучшить представление данных.

	A	C	D	E	F	G
1	Товары в заказе	Отсутствует?	В наличии?	XL2007+	Используйте ЕСЛИ для любых слов	Для вычислений
2	CR50-6	ИСТИНА	ЛОЖЬ	Out of Stock!	Out of Stock - Back Ordered	0
3	BR15-3	ЛОЖЬ	ИСТИНА	BR15-3	Cha-Ching!	1
4	ER49-20	ИСТИНА	ЛОЖЬ	Out of Stock!	Out of Stock - Back Ordered	0
5	BR15-3	ЛОЖЬ	ИСТИНА	BR15-3	Cha-Ching!	1
6	BG33-17	ЛОЖЬ	ИСТИНА	BG33-17	Cha-Ching!	1
7						
8			Использованные формулы:			
9	Товары на ск	C2:	=ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ))			
10	BG33-17	D2:	=НЕ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ)))			
11	BG33-3	E2:	=ЕСЛИОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ);"Out of Stock!")			
12	BR15-3	F2:	=ЕСЛИ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ)); "Out of Stock - Back Ordered";"Cha-Ching!")			
13	CR50-2	G2:	=ЕСЛИ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ));0;1)			
14	Cross50-5					
15	FR 46-33					

Рис. 4.4. Формула с ВПР может возвращать значения ИСТИНА/ЛОЖЬ или любые слова

В столбце С, функция ЕОШИБКА получает результат от ВПР и преобразует его в значение ИСТИНА или ЛОЖЬ. Немного странно, что отсутствие товара дает значение ИСТИНА. Ситуация прояснится, если вы обратите внимание на заголовок в С1 – «Отсутствует?». Формула в С2: =ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ)).

Вполне вероятно, что перевернутая логика столбца С может вас раздражать. Если позвонил клиент и спросил, есть ли некий товар в наличии, вы должны были бы сказать: «это ложь, что товар есть в наличии». Большинство людей так не говорят, поэтому желательно, чтобы значение ИСТИНА означало бы, что товар есть в наличии. Добавьте функцию НЕ перед формулой. Это функция просто поменяет ЛОЖЬ на ИСТИНА и наоборот. Формула в D2 =НЕ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ))).

Начиная с версии 2007 в Excel появились новые функции проверки ошибок, в частности ЕСЛИОШИБКА. Формула в столбце E вернет слова Out of Stock! (Нет на складе). Я не большой поклонник такого подхода. Если вашему менеджеру не понравился столбец В, он не полюбит и столбец E. Формула в E2 =ЕСЛИОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ);"Out of Stock!").

Для абсолютного контроля над тем, какие слова или словосочетания появятся в ответ на запрос, используйте функцию ЕСЛИ. Вы можете указать слова в кавычках, которые будут использоваться для значений ИСТИНА и ЛОЖЬ. Формула в F2 =ЕСЛИ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ));"Out of Stock - Back Ordered";"Cha-Ching!").

Обратите внимание, что первый параметр такой же, как и в формуле C2. Таким образом, значение ИСТИНА соответствует отсутствию товара на складе и первыми надо указать слова, появляющиеся в этой ситуации. Хотя эта формула самая длинная из представленных, именно она позволит вам использовать словосочетания, приятные глазу вашего менеджера.

Что делать, если вам потребуются некоторые расчеты, после того, как вы узнаете, какие товары есть в наличии? Рекомендую воспользоваться тем же типом формулы, что и в F2, но вместо слов использовать 1 для наличия на складе и 0 – для отсутствия. Формула в G2 =ЕСЛИ(ЕОШИБКА(ВПР(A2;\$A\$10:\$A\$23;1;ЛОЖЬ));0;1). В этом случае вы можете умножить цены или штуки на значение 1 или 0 и оценить, какую часть заказа можно заполнить прямо сейчас (рис. 4.5). Хотя общая стоимость заказа \$13 357, в наличии на складе товаров лишь на \$7115 (по-хорошему, еще не хватает оценка наличия заказанного количества, а не только конкретных товаров).

	A	B	C	D	E	F
1	Товары в заказе	Количество	Цена	Всего	В наличии	Всего в наличии
2	CR50-6	50	44,95	\$2 247,50	0	\$0,00
3	BR15-3	75	44,95	\$3 371,25	1	\$3 371,25
4	ER49-20	100	39,95	\$3 995,00	0	\$0,00
5	BR15-3	75	29,95	\$2 246,25	1	\$2 246,25
6	BG33-17	50	29,95	\$1 497,50	1	\$1 497,50
7			Всего	\$13 357,50	Со склада	\$7 115,00
8						
9	Товары на складе					
10	BG33-17					
11	BG33-3					
12	BR15-3					
13	CR50-2					
14	Cross50-5					
15	FR 46-13					

Рис. 4.5. ВПР возвращает нули и единицы, что облегчает вычисления в столбце F

Объединение двух списков. Проблема: у меня есть список продаж по клиентам с начала месяца. Мой коллега только что прислал мне список продаж за вчерашний день. Мне нужно совместить и объединить эти списки.

	A	B	C	D	E
1	Клиент	Месячные продажи		Клиент	Новые продажи
2	AIG	\$210 000		AT&T	\$12 543
3	AT&T	\$1 380 000		CitiGroup	\$98 444
4	Bank of America	\$1 550 000		Exxon	\$143 555
5	Boeing	\$640 000		General Electric	\$3 891
6	Chevron	\$470 000		Home Depot	\$115 433
7	Compaq	\$970 000		HP	\$80 123
8	Duke Energy	\$800 000		Sears	\$125 000
9	Exxon	\$1 570 000			
10	Ford	\$1 930 000			
11	General Motors	\$1 890 000			
12	HP	\$1 450 000			
13	IBM	\$1 460 000			
14	Lucent	\$1 630 000			
15	SBC Communications	\$1 330 000			
16	Sears	\$1 140 000			
17	State Farm	\$1 460 000			
18	Verizon	\$830 000			
19	Wal-Mart	\$1 780 000			
20					

Рис. 4.6 Необходимо объединить эти списки

Задачу можно выполнить в три этапа (рис. 4.7):

- Сначала с помощью ВПР определите, появились ли новые клиенты (есть в таблице 2, но нет в таблице 1). Отсортируйте таблицу 2, собрав все #Н/Д вместе. Формула в F2 =ВПР(D2;\$A\$1:\$B\$19;1;ЛОЖЬ).
- Добавьте новых клиентов в таблицу 1. Теперь таблица 1 включает всех клиентов, но не все продажи. Создайте дополнительный столбец в таблице 1, в который с помощью ВПР поместите новые продажи из таблицы 2. Используйте функцию ЕСЛИОШИБКА, чтобы вместо #Н/Д отразить нули. Формула в C2 =ЕСЛИОШИБКА(ВПР(A2;\$D\$1:\$E\$8;2;ЛОЖЬ);0).
- Создайте еще один столбец в таблице 1, в котором просуммируйте месячные и новые продажи. Формула в D2 =B2+C2.

	A	B	C	D	E	F
1	Клиент	Месячные продажи		Клиент	Новые продажи	Новые
2	AIG	\$210 000		AT&T	\$12 543	AT&T
3	AT&T	\$1 380 000		Exxon	\$143 555	Exxon
4	Bank of America	\$1 550 000		HP	\$80 123	HP
5	Boeing	\$640 000		Sears	\$125 000	Sears
6	Chevron	\$470 000		CitiGroup	\$98 444	#Н/Д
7	Compaq	\$970 000		General Electric	\$3 891	#Н/Д
8	Duke Energy	\$800 000		Home Depot	\$115 433	#Н/Д
9	Exxon	\$1 570 000				
10	Ford	\$1 930 000				
11	General Motors	\$1 890 000				
12	HP	\$1 450 000				
13	IBM	\$1 460 000				
14	Lucent	\$1 630 000				
15	SBC Communications	\$1 330 000				
16	Sears	\$1 140 000				
17	State Farm	\$1 460 000				
18	Verizon	\$830 000				
19	Wal-Mart	\$1 780 000				
20						

	A	B	C	D	E	F	G
1	Клиент	Месячные п	Новые продажи		Клиент	Новые прод	Новые
2	AIG	\$210 000	\$0		AT&T	\$12 543	AT&T
3	AT&T	\$1 380 000	\$12 543		Exxon	\$143 555	Exxon
4	Bank of America	\$1 550 000	\$0		HP	\$80 123	HP
5	Boeing	\$640 000	\$0		Sears	\$125 000	Sears
6	Chevron	\$470 000	\$0		CitiGroup	\$98 444	CitiGroup
7	Compaq	\$970 000	\$0		General Electric	\$3 891	General Ele
8	Duke Energy	\$800 000	\$0		Home Depot	\$115 433	Home Depo
9	Exxon	\$1 570 000	\$143 555				
10	Ford	\$1 930 000	\$0				
11	General Motors	\$1 890 000	\$0				
12	HP	\$1 450 000	\$80 123				
13	IBM	\$1 460 000	\$0				
14	Lucent	\$1 630 000	\$0				
15	SBC Communica	\$1 330 000	\$0				
16	Sears	\$1 140 000	\$125 000				
17	State Farm	\$1 460 000	\$0				
18	Verizon	\$830 000	\$0				
19	Wal-Mart	\$1 780 000	\$0				
20	CitiGroup		\$98 444				
21	General Electric		\$3 891				
22	Home Depot		\$115 433				
23							
24							

	A	B	C	D	E	F	G	H
1	Клиент	Месячные п	Новые пр	Итого		Клиент	Новые пр	Новые
2	AIG	\$210 000	\$0	\$210 000		AT&T	\$12 543	AT&T
3	AT&T	\$1 380 000	\$12 543	\$1 392 543		Exxon	\$143 555	Exxon
4	Bank of Americ	\$1 550 000	\$0	\$1 550 000		HP	\$80 123	HP
5	Boeing	\$640 000	\$0	\$640 000		Sears	\$125 000	Sears
6	Chevron	\$470 000	\$0	\$470 000		CitiGroup	\$98 444	CitiGroup
7	Compaq	\$970 000	\$0	\$970 000		General Electric	\$3 891	General Ele
8	Duke Energy	\$800 000	\$0	\$800 000		Home Depot	\$115 433	Home Depo
9	Exxon	\$1 570 000	\$143 555	\$1 713 555				
10	Ford	\$1 930 000	\$0	\$1 930 000				
11	General Motors	\$1 890 000	\$0	\$1 890 000				
12	HP	\$1 450 000	\$80 123	\$1 530 123				
13	IBM	\$1 460 000	\$0	\$1 460 000				
14	Lucent	\$1 630 000	\$0	\$1 630 000				
15	SBC Communic	\$1 330 000	\$0	\$1 330 000				
16	Sears	\$1 140 000	\$125 000	\$1 265 000				
17	State Farm	\$1 460 000	\$0	\$1 460 000				
18	Verizon	\$830 000	\$0	\$830 000				
19	Wal-Mart	\$1 780 000	\$0	\$1 780 000				
20	CitiGroup		\$98 444	\$98 444				
21	General Electric		\$3 891	\$3 891				
22	Home Depot		\$115 433	\$115 433				
23								

Рис. 4.7. Три этапа объединения данных: а) новые клиенты; б) новые продажи по всем клиентам; в) суммарные продажи по всем клиентам

Если таблица 2 и промежуточные столбцы В и С в таблице 1 вам не нужны (рис. 4.7в), скопируйте данные из D2:D22 в буфер, и вставьте их на то же место специальной вставкой (только значения). После этого таблицу 2 и столбцы В и С можно удалить.

Сортировка с помощью формулы. Проблема: у меня есть список вчерашних продаж торговых представителей. Я хочу использовать формулу, чтобы отобразить результаты нескольких лучших представителей.

	А	В	С
1	Вчерашние продажи		
2			
3		Имя	Продажи
4		Abby	1633
5		Bo	2300
6		Chuck	1931
7		Dale	2443
8		Eddy	1745
9		Flo	1581
10		Gail	2348
11		Helen	2665
12		Ike	1501
13		Josh	2192

Рис. 4.8. Требуется упорядочить этот список, используя формулу

Стратегия: используйте функцию РАНГ, чтобы ранжировать результаты от лучшего к худшему. На рис. 4.9 формула в А4 =РАНГ(С4;С\$4:С\$13).

ВПП : X ✓ fx =РАНГ(С4;С\$4:С\$13)

	А	В	С	Д	Е	F	G
1	Вчерашние продажи						
2							
3	Ранг	Имя	Продажи				
4	=РАНГ(С4;С\$4:С\$13)	Abby	1633				
5	4	Bo	2300				
6	6	Chuck	1931				
7	2	Dale	2443				
8	7	Eddy	1745				
9	9	Flo	1581				
10	3	Gail	2348				
11	1	Helen	2665				
12	10	Ike	1501				
13	5	Josh	2192				
14							

Рис. 4.9. Helen имела вчера наибольший объем продаж, поэтому ее ранг равен 1

Мы создали удобную структуру для извлечения имен на основе ранга. Следующая формула вернет имя лучшего продавца =ВПР(1,\$A\$4:\$C\$13,2,ЛОЖЬ). Однако, эта формула не универсальна и при протаскивании потребуются руками менять 1 на 2, на 3 и т.д. Вместо этого в качестве первого аргумента функции ВПР можно использовать функцию СТРОКА(A1). Эта функция вернет 1, а при протаскивании вниз по столбцу, автоматически перейдет в СТРОКА(A2), которая вернет двойку и т.д. Более того, чтобы в таблице не просто вывести имена лучших торговых представителей, но и объемы продаж, в качестве третьего аргумента функции ВПР (у нас это двойка) можно использовать функцию СТОЛБЕЦ(B1), возвращающую двойку. Тогда при протаскивании вправо по строке, функция автоматически перейдет в СТОЛБЕЦ(C1), возвращающую тройку. Итак, функция в ячейке B19 (рис. 4.10) =ВПР(СТРОКА(A1);\$A\$4:\$C\$13;СТОЛБЕЦ(B1);ЛОЖЬ). Если протянуть функцию по области B19:C23, получим таблицу на рис. 4.5.

B19		=ВПР(СТРОКА(A1);\$A\$4:\$C\$13;СТОЛБЕЦ(B1);ЛОЖЬ)	
A	B	C	D
16	Лучшие продавцы по итогам вчерашних продаж		
17			
18	Имя	Продажи	
19	Helen	2665	
20	Dale	2443	
21	Gail	2348	
22	Bo	2300	
23	Josh	2192	

Рис. 4.10 Формула ВПР фактически сортирует данные из исходной таблицы

К сожалению, функция РАНГ не справится с ситуацией, если двое (или более) торговых представителей будут иметь одинаковый объем продаж (рис. 4.11). Gail и Helen имеют ранг 1, но нет никого с рангом 2. ВПР вернет ошибку при попытке найти представителя с рангом 2.

A	B	C	D	E
1	Равенство объемов продаж вызывает проблемы			
2				
3	Ранг	Имя	Продажи	
4	9	Abby	1501	
5	3	Bo	2300	
6	6	Chuck	1931	
7	3	Dale	2300	
8	8	Eddy	1745	
9	6	Flo	1931	
10	1	Gail	2665	
11	1	Helen	2665	
12	9	Ike	1501	
13	5	Josh	2192	
14				
15	Нет рангов 2 и 4			
16				
17	Имя	Продажи		
18	Gail	2665		
19	#Н/Д	#Н/Д		
20	Bo	2300		
21	#Н/Д	#Н/Д		
22	Josh	2192		

Рис. 4.11. Две пары торговых представителя имеют одинаковый объем продаж и ранг 1 и 3

Пожалуй, лучшее решение – это добавление в формулу, вычисляющую ранг, слагаемое, которое подсчитывает, сколько записей выше этой записи имеют точно такое же значение. На рис. 4.12 формула в A4 примет вид =РАНГ(C4;\$C\$4:\$C\$13)+СЧЁТЕСЛИ(C\$3:C3;C4).

Используйте ВПР + СЧЁТЕСЛИ		
Ранг	Имя	Продажи
9	Abby	1501
3	Bo	2300
6	Chuck	1931
4	Dale	2300
8	Eddy	1745
7	Flo	1931
1	Gail	2665
2	Helen	2665
10	Ike	1501
5	Josh	2192

Дубли отражаются корректно		
Имя	Продажи	
Gail	2665	
Helen	2665	
Bo	2300	
Dale	2300	
Josh	2192	

Рис. 4.12. Модифицированная формула ранжирования справилась с дублями

Еще два слова о формуле в A4. Вы хотите узнать, есть ли в любой клетке выше C4 такой же объем продаж, как и в C4. Конечно, нет, потому что единственная клетка выше C4 является заголовком. Тем не менее, вы хотите ввести эту формулу правильно, чтобы подготовиться к протаскиванию. Вы используете смешанную ссылку =СЧЁТЕСЛИ(C\$3:C3;C4) (подробнее см. [Относительные, абсолютные и смешанные ссылки на ячейки в Excel](#)). В первом аргументе этой функции C\$3 говорит о том, что диапазон сравнения всегда будет начинаться с C3. C3 говорит, что формула будет расширять диапазон сравнения по мере протаскивания вниз. C4 говорит, что вы сравниваете значения в диапазоне со значением в текущей строке. Когда вы копируете формулу вниз на одну строку формула в A5 станет =РАНГ(C5;\$C\$4:\$C\$13)+СЧЁТЕСЛИ(C\$3:C4;C5).